Supplementary materials for:

Tang TS, Funnell MM, Sinco B, Spencer MS, Heisler M. Peer-led, empowerment-based approach to self-management efforts in diabetes (PLEASED): a randomized controlled trial in an African-American community. *Ann Fam Med*. 2015;13:S27-S35. Doi: 10.1370/afm.1819.

Appendix: Using the Multivariate Delta Method to Report Logistic Regression Results with Repeated Measures

By Brandy R. Sinco, MS, University of Michigan, Ann Arbor, MI

Generalized Estimating Equation With Repeated Measures

Note: This example uses time points, but could easily be extended to the four time points in this paper.

- Let i = 0 for control and 1 for treatment.
- Let j = 1 for pre-intervention and 2 for post-intervention.
- Let k = k^{th} subject.
- Let R = 0 for control and 1 for treatment.
- Let T = 0 for baseline and 1 for first follow-up.
- Let π = probability of success.

Generalized Estimating Equation for a Binary Outcome.
$\text{logit}(\pi_{ijk}) = \ln(\pi_{ijk}/(1 - \pi_{ijk})) = (\beta_0 + \beta_1 R + \beta_2 T + \beta_3 RT + \varepsilon_{ijk}$, ε_{ijk} = error term.
The β terms are assumed to be multivariate normal.

Estimated Mean: $\ln(\pi_{ijk}/(1 - \pi_{ijk})) = \beta_0 + \beta_1 R + \beta_2 T + \beta_3 RT$.
The mean percentages for the control group are $\exp(\beta_0)/(1+ \exp(\beta_0))$ at pre-intervention and $\exp(\beta_0 + \beta_2)/(1+ \exp(\beta_0 + \beta_2))$ at post-intervention.
The means for the treatment group are $\exp(\beta_0 + \beta_1)/(1+ \exp(\beta_0 + \beta_1))$ at pre-intervention and $\exp(\beta_0 + \beta_1 + \beta_2 + \beta_3)/(1+ \exp(\beta_0 + \beta_1 + \beta_2 + \beta_3))$ at post-intervention.

The Univariate Delta Method[1]

- Let Y ~ Normal(μ, σ^2), μ, $\sigma^2 \neq 0$; n = sample size.
- Let g(Y) be a differentiable function of Y with non-zero first derivative.
- Then, a first-order Taylor series for g(Y) = g(μ) + g'(μ)(Y – μ).
- Mean of g(Y) ≈ g(μ) and Variance of (g(Y)) ≈ $(g'(\mu))^2\sigma^2$.
- Delta Method Theorem: $\lim_{n\to\infty}\sqrt{n}\left(g(Y)-g(\mu)\right)\xrightarrow{D} Normal\left(0,\sigma^2\left(g'(\mu)\right)^2\right)$.
- I.E., asymptotic distribution of g(Y) = Normal(g(μ), $(g'(\mu))^2\sigma^2$) .

- **Example: Let Y ~ Normal(μ, σ^2). Let W = g(Y) = e^Y.**
- $g'(\mu) = e^\mu$.
- Using the delta method, mean of W = g(μ) = e^μ and
- Variance of W = $(g'(\mu))^2\sigma^2 = e^{2\mu}\sigma^2$; Standard deviation of W = $e^\mu\sigma$.
- Asymptotic distribution of W = Normal(e^μ, $e^{2\mu}\sigma^2$).

The Multivariate Delta Method[2]

- Let Y be a multivariate vector of m normal variables, Y = [Y_1 Y_2 … Y_m].
- Y ~ N(μ, ∑), where ∑ is a m × m covariance matrix.
- Let g(Y) be a differentiable function of Y with non-zero first derivative.

- The multivariate delta method states that if $\lim_{n\to\infty}\sqrt{n}\left(Y-\mu\right)\xrightarrow{D}N\left(0,\Sigma\right)$, then $\lim_{n\to\infty}\sqrt{n}\left(g\left(Y\right)-g\left(\mu\right)\right)\xrightarrow{D}N\left(0,J_g\left(\mu\right)\Sigma J_g^T\left(\mu\right)\right)$.
- Where $J_g(\mu)$ is the Jacobian matrix, evaluated at Y = μ.
- $J_g(\mu) = \begin{bmatrix} \frac{\partial g_1(Y)}{\partial Y_1} & \frac{\partial g_1(Y)}{\partial Y_2} & \cdots & \frac{\partial g_1(Y)}{\partial Y_m} \\ \frac{\partial g_2(Y)}{\partial Y_1} & \frac{\partial g_2(Y)}{\partial Y_2} & \cdots & \frac{\partial g_2(Y)}{\partial Y_m} \\ \vdots & \vdots & \ddots & \vdots \\ \frac{\partial g_m(Y)}{\partial Y_1} & \frac{\partial g_m(Y)}{\partial Y_2} & \cdots & \frac{\partial g_m(Y)}{\partial Y_m} \end{bmatrix}$ evaluated at Y = μ.

Application of the Multivariate Delta Method to Report the Outcomes of a Generalize Estimating Equation for a Binary Variable As Percentages Instead of As Odds Ratios[2].
(Note: Reference shows how to apply the delta method to the log transform when used with a linear mixed model. The same methodology can be used to calculate percentages from a logistic model.)

- The β's are assumed to be multivariate normal, with covariance matrix $\sum_\beta$.
- Percent success of control group at time 1 = $\exp(\beta_0)/(1+\exp(\beta_0))$.
- $g_1(\beta) = \exp(\beta_0)/(1+\exp(\beta_0))$.
- Note that $\partial g_1(\beta)/\partial\beta_0 = \exp(\beta_0)/(1+\exp(\beta_0))^2$.and $\partial g_1(\beta)/\partial\beta_i = 0$ if i>0.
-
- Percent success of control group at time 2 = $\exp(\beta_0+\beta_2)/(1+\exp(\beta_0+\beta_2))$.
- Let $g_2(\beta) = \exp(\beta_0+\beta_2)/(1+\exp(\beta_0+\beta_2))$. Note that $\partial g_2(\beta)/\partial\beta_0 = \partial g_2(\beta)/\partial\beta_2 = \exp(\beta_0+\beta_2)/(1+\exp(\beta_0+\beta_2))^2$ and $\partial g_2(\beta)/\partial\beta_i = 0$ for $i\neq 0$ and $i\neq 2$.

- Same property holds for g_3 and g_4 derivatives with respect to the β's.
- Percent success of treatment group at time 1 = $\exp(\beta_0+\beta_1)/(1+\exp(\beta_0+\beta_1))$.
- $g_3(\beta) = \exp(\beta_0+\beta_1)/(1+\exp(\beta_0+\beta_1))$.
- $\partial g_3(\beta)/\partial\beta_0 = \partial g_3(\beta)/\partial\beta_1 = \exp(\beta_0+\beta_1)/(1+\exp(\beta_0+\beta_1))^2$, and $\partial g_3(\beta)/\partial\beta_i = 0$ for $i>1$.
-
- Percent success of treatment group at time 2 = $\exp(\beta_0+\beta_1+\beta_2+\beta_3)/(1+\exp(\beta_0+\beta_1+\beta_2+\beta_3))$.
- $g_4(\beta) = \exp(\beta_0+\beta_1+\beta_2+\beta_3)/(1+\exp(\beta_0+\beta_1+\beta_2+\beta_3))$
- $\partial g_4(\beta)/\partial\beta_i = \exp(\beta_0+\beta_1+\beta_2+\beta_3)/(1+\exp(\beta_0+\beta_1+\beta_2+\beta_3))^2$, $0 \le i \le 3$.

Jacobian Matrix for g(β).

$$J_g(\beta)=\begin{bmatrix} \frac{\exp(\beta_0)}{\left(1+\exp(\beta_0)\right)^2} & 0 & 0 & 0 \\ \frac{\exp(\beta_0+\beta_2)}{\left(1+\exp(\beta_0+\beta_2)\right)^2} & 0 & \frac{\exp(\beta_0+\beta_2)}{\left(1+\exp(\beta_0+\beta_2)\right)^2} & 0 \\ \frac{\exp(\beta_0+\beta_1)}{\left(1+\exp(\beta_0+\beta_1)\right)^2} & \frac{\exp(\beta_0+\beta_1)}{\left(1+\exp(\beta_0+\beta_1)\right)^2} & 0 & 0 \\ \frac{\exp\left(\sum_{i=0}^{3}\beta_i\right)}{\left(1+\exp\left(\sum_{i=0}^{3}\beta_i\right)\right)^2} & \frac{\exp\left(\sum_{i=0}^{3}\beta_i\right)}{\left(1+\exp\left(\sum_{i=0}^{3}\beta_i\right)\right)^2} & \frac{\exp\left(\sum_{i=0}^{3}\beta_i\right)}{\left(1+\exp\left(\sum_{i=0}^{3}\beta_i\right)\right)^2} & \frac{\exp\left(\sum_{i=0}^{3}\beta_i\right)}{\left(1+\exp\left(\sum_{i=0}^{3}\beta_i\right)\right)^2} \end{bmatrix}$$

In simpler form, $J_g(\beta)=\begin{bmatrix} W_1 & 0 & 0 & 0 \\ W_2 & 0 & W_2 & 0 \\ W_3 & W_3 & 0 & 0 \\ W_4 & W_4 & W_4 & W_4 \end{bmatrix}$.

Define the covariance matrix for the β's as $\Sigma_\beta=\begin{bmatrix} \sigma^2_{\beta 00} & \sigma_{\beta 01} & \sigma_{\beta 02} & \sigma_{\beta 03} \\ \sigma_{\beta 01} & \sigma^2_{\beta 11} & \sigma_{\beta 12} & \sigma_{\beta 13} \\ \sigma_{\beta 02} & \sigma_{\beta 12} & \sigma^2_{\beta 22} & \sigma_{\beta 23} \\ \sigma_{\beta 03} & \sigma_{\beta 13} & \sigma_{\beta 23} & \sigma^2_{\beta 33} \end{bmatrix}$.

The covariance matrix for W (estimates for each group at each time point) will be $\sum_W = J_g(\beta) \times \sum_\beta \times (J_g(\beta))^T$.

Example SAS Code to Compute Covariance Matrix With Four Time Points (Other software could be used. The delta method can be applied with any statistical software.)

```
*** PHQ Bin, Minimally Depressed or Greater ***;
ods html; ods graphics on;
Proc Genmod Data=AcrossTimeLong_Ypsi Descending;
Class REACHID TimepointN RandomizationN;
Model PHQBin=TimepointN RandomizationN TimepointN*RandomizationN / dist=bin;
Repeated Subject=REACHID / Type=UN Modelse ECOVB;
ODS OUTPUT GEEModPEst=RegBinPHQ GEERCov=CovBPHQ;
Run;
```

```
ods graphics off; ods html close;

/* PHQBin: Compute standard errors for difference scores via the delta method */
ods html path="c:\temp";
Proc IML;
/* Input beta vector; begin at B1 instead of B0 because SAS numbers parameters
starting at 1 */
use regbinphq; read all var {'Estimate'} into B; close regbinphq;
/* B5=B7=B9=B11=B13=B14=B15=0 reference levels */
B00 = B[1];   /* B00 = control group at baseline */
B10 = B[1] +  B[6];   /* B10 = peer group at baseline */
B01 = B[1] + B[4];  /* B01 = control group at 3 months */
B11 = B[1] + B[4] + B[6] + B[12]; /* B11 = peer group at 3 months */
B02 = B[1] + B[3];  /* B02 = control group at 9 months */
B12 = B[1] + B[3] + B[6] + B[10]; /* B12 = peer group at 9 months */
B03 = B[1] + B[2];  /* B03 = control group at 15 months */
B13 = B[1] + B[2] + B[6] + B[8];  /* B12 = peer group at 15 months */

/* Convert to exp scale */
expB00=exp(B00)/((1+exp(B00))**2);
expB10=exp(B10)/((1+exp(B10))**2);
expB01=exp(B01)/((1+exp(B01))**2);
expB11=exp(B11)/((1+exp(B11))**2);
expB02=exp(B02)/((1+exp(B02))**2);
expB12=exp(B12)/((1+exp(B12))**2);
expB03=exp(B03)/((1+exp(B03))**2);
expB13=exp(B13)/((1+exp(B13))**2);

/* Input covariance matrix, 8x8 matrix */
use CovBPHQ; read all var{'Prm1' 'Prm2' 'Prm3' 'Prm4' 'Prm6' 'Prm8' 'Prm10' 'Prm12'}
into CovB; close CovBPHQ; print CovB;

/* Contruct W, 8x8 matrix */
W=j(8,8,0);
W[1,1]=expB00;

z={1 5};
W[2,z]=expB10;

z={1 4};
W[3, z]=expB01;

z={1 4 5 8};
W[4,z]=expB11;

z={1 3};
```

```
W[5,z]=expB02;

z={1 3 5 7};
W[6,z]=expB12;

W[7,1:2]=expB03;

z={1 2 5 6};
W[8,z]=expB13;

print W;

/* V=W*CovB*T(W) = cov matrix of exp(Bij)/((1+exp(Bij))**2), i=0 to 1, j=0 to 3 */
reset fuzz; /* This corrects rounding errors. */
V=W*CovB*T(W);
print V;
quit;
ods html close;
```

REFERENCES

[1]Multivariate Delta Method: Reference: Casella G, Berger RL. Statistical inference. 2nd ed. Pacific Grove, CA: Duxbury; 2002.

[2]Sinco B., Kieffer E., Spencer M. Using the Delta Method to generate means and confidence intervals from a Linear Mixed Model on the original scale, when the analysis is done on the log scale. Presented at the American Statistical Association's Conference on Statistical Practice in New Orleans, 2/20/2015.